Integration in polar coordinates

Worksheet by Mike May, S.J.- maymk@slu.edu

```
> restart:with(plots):
```

Start by opening the section below.

A review of plotting in polar coordinates:

The first problem in trying to do double integrals in polar coordinates is to be able to sketch graphs of functions described in polar coordinates. On your calculators you switch to polar mode. With Maple you use the coords=polar option on the plot command. You can either plot the curve either with r as a function of theta, or with both r and theta described as functions of a parameter t. When we plot in polar coordinates it is probably wise to use the scaling=CONSTRAINED option so the axes have the same scale.

```
> plot(cos(2*theta),theta=0..2*Pi, coords=polar, scaling=
  CONSTRAINED);
  plot([2+sin(2*t),Pi*sin(t),t=0..Pi], coords=polar, scaling=
  CONSTRAINED);
```





To plot several curves together at the same time we use set notation, just like we did in Cartesian coordinates.

```
> plot({1,2*sin(theta)}, theta=0..2*Pi, coords=polar, scaling=
  CONSTRAINED);
```



Plotting several curves together lets us plot regions made of a number of curves. The region plotted below is the portion of an annulus between two specified angles.

```
> plot({[1+t,Pi/6,t=0..1],[1+t,Pi/3,t=0..1],[1,t,t=Pi/6..Pi/3],
     [2,t,t=Pi/6..Pi/3],[0,0,t=0..1]},coords=polar, scaling=
  CONSTRAINED);
```



We can also use the implicitplot command from the plots package when we want to plot a number of curves, some in terms of r and some in terms of theta.

```
> implicitplot([r=1, r=2, theta=Pi/6, theta=2*Pi/3, r*cos
  (theta)=1.5],
    r=0..3, theta=0..2*Pi, color=[red, green, blue, yellow,
  brown],
    coords=polar);
```



Exercise:

1) Plot the curves $r = 3\cos(\theta)$ and $r = 1 + \cos(\theta)$. Find the points of intersection. (You may want to use either your calculator or a piece of paper to find the points of intersection.)

```
>
```

Once we can sketch curves, the problems involved in setting up an integral in polar coordinates are similar to the problems involved in setting up a double integral in Cartesian coordinates. The biggest problem is finding the correct limits of integration. We will also be concerned with switching the order of integration.

Finding the limits of integration:

Setting up drdθ integrals

For our first example we want to consider the region bounded by the curves
theta=Pi/3, theta=3*Pi/2, r=1.5-sin(theta), and r=3+cos(theta).
As with the Cartesian case, we start by plotting the curves implicitly to see what is going on.

```
> implicitplot([r=1.5-sin(theta), r=3+cos(theta), theta=Pi/6,
  theta=3*Pi/2],
    r=0..4, theta=0..2*Pi, coords=polar, axes=boxed,
    color=[red, green, blue, yellow]);
```



Given the picture, we have a region bounded by curves $r = g(\theta)$ and $r = h(\theta)$ with the value of θ bound by two constants. Thus we want to integrate in r first, using dr$d\theta$.
We check that the integral sets up correctly.

```
> r:='r':  theta := 'theta':
  lowtheta  := Pi/6;
  hightheta := 3*Pi/2;
  lowr  := theta -> 1.5-sin(theta);
  highr := theta -> 3 + cos(theta);
  print(`Region of integration for `,
       Int(Int(f(r, theta)*r,r=lowr(theta)..highr(theta)),
           theta=lowtheta..hightheta));
```

$$lowtheta := \frac{1}{6}\pi$$

$$hightheta := \frac{3}{2}\pi$$

$$lowr := \theta \rightarrow 1.5 - \sin(\theta)$$

$$highr := \theta \rightarrow 3 + \cos(\theta)$$

(2.1.1)

$$\textit{Region of integration for}, \int_{\frac{1}{6}\pi}^{\frac{3}{2}\pi} \int_{1.5-\sin(\theta)}^{3+\cos(\theta)} f(r,\theta)\, r \,\mathrm{d}r\,\mathrm{d}\theta \qquad \textbf{(2.1.1)}$$

The integration with respect to r for a particular θ integrates along a radial line. The following block of code is designed to help you visualize what the limits of integration mean.

```
> r:='r':  theta := 'theta':
  lowtheta  := Pi/3;
  hightheta := 3*Pi/2;
  lowr  := theta -> 1;
  highr := theta -> 3 + cos(theta);
  print(`Region of integration for `,
       Int(Int(f(r, theta)*r,r=lowr(theta)..highr(theta)),
           theta=lowtheta..hightheta));
  inside := plot([lowr(theta), theta,theta=lowtheta..
  hightheta],
             color=red, coords=polar):
  outside := plot([highr(theta), theta,theta=lowtheta..
  hightheta],
             color=green, coords=polar):
  line := {}:
   for i from 0 to 10 do
      tval := evalf(lowtheta + i/10*(hightheta-lowtheta)):
      if (abs(evalf(lowr(tval) - highr(tval)))>0) then
         line := line union
            {[r,tval, r=evalf(lowr(tval))..evalf(highr(tval))
  ]};
      end if
   end do:
  plotlines := plot(line,coords=polar, color=BLACK) :
  plots[display]({inside, outside, plotlines},scaling=
  CONSTRAINED);
```

$$lowtheta := \frac{1}{3}\pi$$

$$hightheta := \frac{3}{2}\pi$$

$$lowr := \theta \rightarrow 1$$

$$highr := \theta \rightarrow 3+\cos(\theta)$$

$$\textit{Region of integration for}, \int_{\frac{1}{3}\pi}^{\frac{3}{2}\pi} \int_{1}^{3+\cos(\theta)} f(r,\theta)\, r \,\mathrm{d}r\,\mathrm{d}\theta$$



Note that since we integrate with respect to r first, the r-limits are functions of theta while the theta limits are constants. We integrate by first integrating, from the inside (red) curve to the outside (green) curve, along the radial lines.

Exercises:

2) Find the limits of integration to integrate over the region inside the curve $r = 1 + \cos(\theta)$ and outside the curve r=1. Modify the code above to show that you have the correct region.

```
>
```

3) Find the limits of integration to integrate over the region inside the curve $r = 3\cos(\theta)$ and outside the curve $r = 1 + \cos(\theta)$. Modify the code above to show that you have the correct region.

```
>
```

Setting up dθdr integrals

For our next example we want the region bounded by the circles of radius 1 and 5, and between the curves theta=Pi*r/6 and theta=Pi*(2-r/12).

```
> implicitplot([r=1, r=5, theta=Pi*r/6, theta=Pi*(2-r/12)],
    r=0..6, theta=0..2*Pi, coords=polar, axes=boxed,
    color=[red, green, blue, yellow]);
```



In this case the region of integration is bounded by curves $\theta = g(r)$ and $\theta = h(r)$ with the value of r being bound by two constants. Instead of integrating first on radial lines, we start by integrating along circular arcs with a fixed value of r. Thus we can set up integrals using drdθ.

Next we check the integral

```
> r:='r':  theta := 'theta':
  lowr  := 1;
  highr := 5;
  lowtheta  := r -> Pi*r/6;
  hightheta := r -> Pi*(2-r/12);
  print(`Region of integration for `, Int(Int(f(r, theta)*r,
               theta=lowtheta(r)..hightheta(r)),r=lowr..
  highr));
```

$$lowr := 1$$

$$highr := 5$$

$$lowtheta := r \rightarrow \frac{1}{6}\pi r$$

$$hightheta := r \rightarrow \pi\left(2 - \frac{1}{12}r\right)$$

$$\text{Region of integration for },\int_{1}^{5}\int_{\frac{1}{6}\pi r}^{\pi\left(2-\frac{1}{12}r\right)} f(r,\theta)\, r\, d\theta\, dr \qquad \textbf{(2.2.1)}$$

Now we check that we have the correct region with curves put in for the inside integral.

```
> r:='r':  theta := 'theta':
  lowr   := 1;
  highr  := 5;
  lowtheta  := r -> Pi*r/6;
  hightheta := r -> Pi*(2-r/12);
  print(`Region of integration for `, Int(Int(f(r, theta)*r,
               theta=lowtheta(r)..hightheta(r)),r=lowr..
  highr));
  lowthetacurve := plot ([r,lowtheta(r), r=lowr..highr],
             color=red, coords=polar) :
  highthetacurve := plot ([r, hightheta(r), r=lowr..highr],
             color=green, coords=polar) :
  arcs := {} :
    for i from 0 to 10 do
                        tempr := evalf(lowr + i/10*(highr-
  lowr)):
     if (abs(evalf(lowtheta(tempr)-hightheta(tempr)))>0) then

         arcs := arcs union {[tempr, theta,
           theta=lowtheta(tempr)..hightheta(tempr)]}:
      end if:
    end do:
  grapharcs := plot(arcs,coords=polar, color=BLACK) :
  plots[display] ( {lowthetacurve, highthetacurve, grapharcs
  }
           ,scaling=CONSTRAINED) ;
```

$$lowr := 1$$

$$highr := 5$$

$$lowtheta := r \rightarrow \frac{1}{6}\pi r$$

$$hightheta := r \rightarrow \pi\left(2-\frac{1}{12}r\right)$$

$$\text{Region of integration for },\int_{1}^{5}\int_{\frac{1}{6}\pi r}^{\pi\left(2-\frac{1}{12}r\right)} f(r,\theta)\, r\, d\theta\, dr$$



Exercises:

4) Find the limits of integration to integrate over the region inside both curves $r = 1 + \cos(\theta)$ and r=1. Modify the code above to show that you have the correct region. Explain why you want to use dthetadr rather than drdtheta for this problem.

>

5) Find the limits of integration to integrate over the region inside the curve $r = 2\sin(\theta)$ and outside the curve $r = \frac{1}{2}$. Modify the code above to show that you have the correct region.

>

Changing order of integration

Some regions can be described to use either drdθ or dθdr. When we switch the order of integration we need to switch the limits as well.

Exercise:

6) The region inside the curve r=1 and outside the curve $r = 1 + \cos(\theta)$ can be set up in either order. Find the limits of integration both ways. Show that you have the correct region.

>

Integrating over a region in polar coordinates:

Recall that dA is rdrdθ or r dθdr. Thus to find the area of the region bounded by r=1, r=2, $\theta = \frac{\pi}{6}$, and $\theta = \frac{\pi}{3}$, we evaluate the integral $\int_{\frac{\pi}{6}}^{\frac{\pi}{3}} \int_1^2 r \, dr \, d\theta$.

```
> Int(Int(r,r=1..2),theta=Pi/6..Pi/3)=int(int(r,r=1..2),theta=Pi/6..Pi/3);
```

$$\int_{\frac{1}{6}\pi}^{\frac{1}{3}\pi} \int_1^2 r \, dr \, d\theta = \frac{1}{4}\pi$$

Exercises:

7) Find the area inside both curves r=1 and $r = 1 + \cos(\theta)$.

```
>
```

8) Integrate the function $\sin(r^2)$ over the disk of radius 2 centered at the origin.

```
>
```

Changing coordinates systems to integrate

One of the reasons we want to be able to integrate in polar coordinates is that some integrals work out nicely in one coordinate system and are ugly in another. To change an integral in Cartesian coordinates into polar coordinates, we need to do several things. First sketch the region with its boundary curves. Then change the formulas of the boundary curves, the function to be integrated and dA into polar form. We are then ready to set up the integral and integrate.

Consider the integral $\int_{-b}^{b} \int_{-\sqrt{b^2 - x^2}}^{\sqrt{b^2 - x^2}} \frac{1}{a^2 + x^2 + y^2} \, dy \, dx$.

```
> int(int(1/(a^2+x^2+y^2),y=-sqrt(b^2-x^2)..sqrt(b^2-x^2)),x=-b..b);
```

$$\int_{-b}^{b} \frac{2 \arctan\left(\frac{\sqrt{b^2 - x^2}}{\sqrt{a^2 + x^2}}\right)}{\sqrt{a^2 + x^2}} \, dx$$

Depending on the version of Maple you are using, it either chokes on this integral, or gives an answer involving functions we don't know how to evaluate. However the integral above converts to $\int_0^b \int_0^{2\pi} \frac{r}{a^2 + r^2} \, d\theta \, dr$ which can easily be done by hand using the substitution $u = r^2$. Maple also has no problem with it.

```
> int(int(r/(a^2+r^2),theta=0..2*Pi),r=0..b);
Warning, unable to determine if -(-a^2)^(1/2) is between 0
and b; try to use assumptions or set _EnvAllSolutions to true
```

```
Warning, unable to determine if (-a^2)^(1/2) is between 0 and
b; try to use assumptions or set _EnvAllSolutions to true
```

$$\int_0^b \frac{2\,r\,\pi}{a^2+r^2}\,dr$$

Maple is still having some problems. Since it is complaining about the square root of - a^2, we want to let Maple know that a is real.

```
> assume(a,real):
  int(int(r/(a^2+r^2),theta=0..2*Pi),r=0..b);
```

$$-\ln\left(a\sim^2\right)\pi+\pi\ln\left(a\sim^2+b^2\right) \tag{4.1}$$

Exercise:

9) Convert the integral $\int_0^{\sqrt{2}}\int_y^{\sqrt{4-y^2}} x\,y\,dx\,dy$ to polar form and evaluate.

\>

Extra credit -

At the beginning of this worksheet we plotted a region described by the parametric curve [2+sin(2*t),Pi*sin(t),t=0..Pi]. I am interested in finding the area of the region bounded by this curve to 3 decimal places. Find the area of the region and cleanly write up your work carefully justifying your method.

\>

\>